
TCL/TK GUIDELINES

1.0 INTRODUCTION

There are two excellent guides on which this coding style is based: the “Tcl
Style Guide” and the “Tcl/Tk Engineering Manual” from
http://resource.tcl.tk/resource/doc/papers/.  The “Tcl Style Guide” focuses on
recommended methods of writing Tcl/Tk scripts, while the “Tcl/Tk Engineering
Manual” discuses methods for linking Tcl to C / C++ code.  Both of these
manuals describe how Tcl is used elsewhere, and it was felt that MDL should
use similar (if not the same) conventions.  The rest of this is a brief
summary of “Tcl Style Guide”, interspersed with MDL specific modifications. 
Following that is an appendix of the current “template.tcl”, the original
“template.tcl, and then a simple example.

1.1 A FEW WORDS ON TCL/TK

Tcl by itself is a powerful scripting tool, which can aid in tasks that are
too complex to do using simple “sh” scripts.  In addition, Tcl provides simple
methods to hook into compiled C code, which in turn allows one to use C,
FORTRAN, C++, or any compiled language to give the program extra speed or
flexibility.  Tcl combined with Tk allows for fast development of Graphical
User Interfaces (GUIs).  Since Tcl/Tk has been ported to all major operating
systems, GUIs created in Tcl/Tk are easily ported, which saves developer time. 
So Tcl/Tk is a good scripting language to initially create proto-types for
programs, and later to be the glue that links the compiled code together.

1.2 WHY CONVENTIONS? 

First, conventions ensure that certain important things get done; for example,
every procedure must have documentation that describes each of its arguments
and its result. Second, the conventions guarantee that all Tcl/Tk code has a
uniform style.  This makes it easier to read and maintain other peoples code. 
Third, the conventions help to avoid some common mistakes by prohibiting
error-prone constructs.

Please write your code to conform to the conventions from the very start. 
Don’t write comment free code, with the intention of going back and putting
them in later.  It won’t happen!

1.3 START-UP SCRIPTS (GETTING THINGS TO RUN)

In every Tcl/Tk project there is one .tcl file that starts everything else. 
To get things to run from an MS-Windows perspective, you simply need to double
click on this script (or make a short-cut to it with a target of
“c:/tcl/bin/wish80.exe <startup script>”.  

From a UNIX perspective, the start-up script is the one that you do the chmod
775 to and it is recommended that the first 3 lines are:
#!/bin/sh
# the next line restarts using wish \
exec wish8.0 “$0" “$@”

In both cases wish8.0 (or wish80.exe) can be replaced with whatever is
evaluating your code, either tclsh (no Tk) or some code that you compiled to
add extra functions to the standard “wish” (Tcl/Tk).  Note the ‘\’ at the end

of the second line is required.  It is important that no blanks or tabs follow
the ‘\’.

1.4 ORGANIZING CODE FILES

Each code file should contain either an entire application, or a set of
related procedures.  Try to keep files in the range of 500-2000 lines.  If
they get larger than that, it becomes difficult to remember what procedures
are where.  If it is smaller, you get too many files which is also difficult
to manage.

2.0 MDL STYLE GUIDE FILE HEADER

#*****************************************************************************
# <filename>
#
#   <Brief Module Description if this is the primary .tcl file>
#   or <Brief file description, mentioning the primary .tcl file>
#
# History:
#   <mm/yyyy> <Programmer> (<Organization>): <Modification comments>
#
# Notes:
#   <Any features that the programmer thinks would be nice, but hasn't
#    implemented yet, also mentionable features.>
#
#*****************************************************************************
# Global variables:         (This section is intended to describe the Global
variables)
#   <Global array name>
#     (<indices>) : <what it contains>
#   <Any other global variables> : <what they contain>
#*****************************************************************************
set src_dir [file dirname [info script]]
if {[file pathtype $src_dir] != "absolute"} {
  set cur_dir [pwd] ;  cd $src_dir
  set src_dir [pwd] ;  cd $cur_dir
}
package require ...
package provide ...
source [file join $src_dir <secondary .tcl files>]
set foobar ...
namespace eval ... {
  source [file join $src_dir <secondary .tcl files>]
  namespace export ...
}

This is similar to the Tcl Guide, except we switched the “copyright” with the
“History” section, and added a “Notes” section.  Also we were more explicit
about the importance of commenting the global variables.  The package
definitions start with the require statements, followed by any source
statements.  Most of the global variables are after the source statements (ie
“set foobar ...”), with the exception of the “src_dir” variable which can be
useful for finding the path to the “secondary” scripts, and for changing the
“auto_path” variable which is used by the package require command.  After the
source statements, we initialize the global variables and set up the name
space definition with the export list first.  When you need to use multiple
files for the same package or logical grouping of code, use a similar format
noting that it is a “secondary” file and mentioning the “primary” file. 

3.0 MDL STYLE GUIDE PROCEDURE HEADER

#*****************************************************************************
# <namespace>::<Procedure name> --
#
#    Brief description (possibly including who should call it and under what
circumstances)
#
# Arguments:
#    <var name 1> : (optional) Defaults to ..., which means ... expected
type... 
#                              what is its function...
#    <var name 2> :  expected type... what is its function...
#
# Globals:
#   <global variables> : What is it used for..., what does it contain.
#
# History:
#   <mm/yyyy> <Programmer> (<Organization>): <Modification comments>
#
# Notes:
#   <Any features that the programmer thinks would be nice, but hasn't
#    implemented yet, also mentionable features.>
#*****************************************************************************
proc <namespace>::<Procedure name> {var1 var2 {var3 1} var4 etc} {
 ...
}

Note 1: For those using namespaces, the procedure is defined outside the
namespace command.

Note 2: Try to sort the arguments by “input” “input/output” “output”,
except as follows:

1) If the argument is actually a sub command for the
command, put it first.

2) If there is a group of procedures all working on an
argument of a particular type, that argument should
either be first, or just after the sub command.

This is very similar to the “Tcl Style Guide”, except we added the Globals,
History, and Notes sections while removing the Results section.  It was felt
that the Results section would be included in the description or the Notes
section so it was not needed.  At one time, it was felt that one could remove
sections if they were not applicable, for example if you didn’t use any global
variables.  This is frowned upon, because it is often easiest to copy and
paste procedure headers, so if a section is missing when it isn’t needed, it
can get “forgotten” when commenting a procedure which needs it.

4.0 LOW LEVEL CODING CONVENTIONS

4.1 INDENTATION
 
Make sure you are consistent. The “Tcl Style Guide” suggests 4 spaces, but I
prefer 2 spaces, since I often run out of space on an 80 column page if I use
4 spaces.  If you are editing someone else’s code, maintain their convention. 
Also, try to avoid the “Tab” as much as possible, since it is inconsistent as
to how many spaces it equals.

I can’t stress enough that indentation is important in Tcl.  This is because
the most annoying bug is the “missing close brace”.  With proper indentation,
you instantly see when you haven’t closed all the braces.  Since Tcl is an
interpreted language, we don’t have the compiler to check for close brace
problems, so we should do it ourselves.

Another place indentation is useful is when creating widgets in Tk.  Since
these tend to be tree oriented, it is a good idea to indent as you develop the
various levels of the tree.  For an example see Section 7.5.

4.2 COMMENTS SHOULD TAKE UP FULL LINES

1) Try to avoid “tacking” on comments to the end of code by using a
;#   This is difficult for others to read, and can cause one to
place the comment where it will cause errors. 

2) Indent the comments to the same level as the surrounding code.

3) Either use ##### or a blank line before and after comments.  This
separates the comments from the code, and makes them easier to
find.  You can omit the proceeding ##### or blank line if the
comment is at the beginning of a block (or indentation level).

4.3 CONTINUATION LINES

Use continuation lines to make sure that no line exceeds 80 columns. 
Continuation lines should be indented to at least twice the normal indentation
level so they aren’t confused with other code.  If you are indenting by 4
characters, the continuation line should be indented to at least 8 characters. 
Try to pick clean places to break your lines for continuation.  One way might
be to start the continuation line with an operator such as * && or ||

4.4: ONE COMMAND PER LINE

Avoid the ; character.  Having one command per line makes code easier to read
and debug.  The only possible exception I can think of would be something like
setting x,y pairs:
set x 5;    set y 21

4.5 ALWAYS USE THE “RETURN” STATEMENT

For procedures that return “void” it doesn’t hurt, and may actually speed up
the code.  For procedures that do return something, it makes it clear what is
being returned and from where.

4.6 IF STATEMENT

Don’t use the “then” command, as it is extraneous.  The else and elseif

commands are useful, and make the code readable and closer to the C language.

4.7 PARENTHESIZE EXPRESSIONS

Use parentheses around each subexpression in an expression to make it
absolutely clear what you mean.  Don’t rely on the precedence rules to solve
things, because all you will do is confuse the next person to look at the
code.
Avoid:
    if {$x > 22 && $y <= 47} ...
Use:
   if {($x > 22) && ($y <= 47)} ...

4.8 CURLY BRACES: { GOES AT THE END OF A LINE

In C / C++ there is a debate as to whether a curly brace should be on the next
line or not. I prefer it with the { at the end of the line in C / C++.  In Tcl
one would have to do:
    for {set i 0} {$i < 20} {incr i} \
    {
       ...
    }
It is very easy to forget the ‘\’ (or inadvertently add a space after the
‘\’), which causes the code to break.  So in Tcl/Tk, simplify your life and
use: 
    for {set i 0} {$i < 20} {incr i} {
       ...
    }

Also, control structures should always have braces, even if there is only one
statement.
Avoid:
    if {$f_test} return.
Instead use:     
    if {$f_test} {
      return
    }

4.9 SWITCH STATEMENT

Always use the -- since it avoids having the string confused with an option. 
Example:
switch -- $command {
  add {
    ...
  }
  default {
    ...
  }
}

4.10 AVOID ‘/’ IN FILENAMES

For platform related issues it is best to use [file join ] so that you can
handle the file separator character better.

5.0 VARIABLES CONSIDERATIONS

5.1 VARIABLE NAMING

1) Be consistent.  Use the same name for the same thing in different
places.  This allows you to copy and paste code, as well as making
the code easier to read.

2) Choose names that make sense.  If someone else sees the name out
of context, do they have a chance of understanding what it is for.

3) Use unique names.  Don’t use “str” for one thing and “string” for
another.  You will get yourself and others confused.

4) Is the name too generic?  Try to choose names that convey some
information.

5) Don’t choose too long a name.  If the name is longer than 80
characters it is probably too long, and you should think about
abreviations.  The other considerations will keep you from having
too short a name.

5.2 VARIABLE SYNTAX RULES

1) “private” variables and procedures should start with an upper case
letter, while “public” variables and procedures should start with
a lower case letter.  By “public” I mean things that people using
the procedure or package should “see”.  Similarly “private” is for
internal use only. Note: this does not necessarily apply to
variables that exist only inside a procedure, but rather for
global variables and namespace variables that others can see.

2) In multi-word names it is recommended that you capitalize the
first letter of each trailing word, and don’t use dashes or
underscores to separate words.  Although underscores are
preferable to dashes.

Example of 1) and 2):
  set PrivateVar 1 
  set publicVar 2

3) Any variable whose value refers to another variable should have a
name ending in “Name”.  It should also indicate what type of
variable it is referring to.

Example: upvar 1 $arrayName array

The following two suggestions apply to infrequent uses, but were added for
completeness:

4) Variables that hold Tcl code to be “evaled” should have names
ending in “Script”.

5) Variables holding a partial Tcl command that must have arguments
appended before becoming a script should end in “Cmd”.

6.0 DOCUMENTING CODE

Documenting is done to save time and reduce errors.  Typically we have two
different purposes for writing documentation.  The first is to teach someone
else how to use our code.  They don’t care how our procedures work, just how
to call them.  The procedure header is the main source for these people.  The
second purpose is to show how it works internally so that someone else is able
to fix bugs or add features easily.  Here good names and indentation will
already make the code readable, but extra notes will allow people to figure
things out faster.

Key Ideas:

1) Document things of wide impact.  Here we mean procedure
interfaces, and global variables.  Thus the emphasis on Procedure
Headers and File Headers in Section 2 and Section 3. 

2) Don’t just repeat what is in the code.  Documentation should
provide higher level information about what is going on.  

3) Don’t repeat your comments.  Document each major design decision
once, and as close as reasonable to where it is implemented.  For
major design decisions it may be difficult to figure out which of
several places to put the comments.  If so, choose one, and
reference it from the other places you might have put it.

4) Keep internal documentation short.  Too much documentation makes
it hard to see the code.  The code should be able to speak for
itself, we just want to enhance and clarify it.  Try to keep major
things about a procedure in the procedure header, and use the
internal documentation to clarify the details.  

5) Write clean code (Keep it simple).  If it takes a lot of
documentation to explain a section of code, you may need to
rethink that code. 

6) Document tricky situations. It isn’t always possible to keep code
simple, so when you feel that the code is tricky or subtle, this
is when you need comments the most.  Spending a little extra time
clarifying the code will save you a lot later.  One particular
instance of this is if you discover subtle properties while fixing
a bug, be sure to add a comment explaining the problem and its
solution.

7) Long procedures need more internal documentation.  You will want
to spend time commenting internal variables, and emphasizing what
the different sections of the procedure are for.  Conversely, for
short procedures you may not need any internal documentation.

8) Document as you go.  Don’t say: “I’ll go back and comment later.” 
Later never comes.

7.0 MISCELLANEOUS

7.1 PACKAGES AND NAMESPACES

As you develop more complex code you will invariably start working with
namespaces and packages.  Packages provide a “standard” method for providing a
set of Tcl/Tk code or C code to someone else.  Namespaces provide a means of
encapsulating your code, and reducing the chances that your procedures and
variables will conflict with someone else’s.  Both are good concepts, but
haven’t been discussed much here as they add complexity and can be
intimidating.  Beginners will not need them, and more advanced programmers
will most likely have a book or two to describe them.  As far as style, the
same ideas for normal procedures and programs hold for both namespaces and
packages.  The one idea to think about is to try to choose a unique name for
your project.  You can see what others have called their packages by checking
with the comp.lang.tcl newsgroup or going to www.scriptics.com and looking
around.  You might even check with NIST Identifier Collaboration Service at
http://pitch.nist.gov/nics

7.2 TEST SUITES

This is another advanced topic, and you probably should read the “Tcl Style
Guide” for the best description of it.  The general idea is that there is a
“test” routine which will evaluate parts of your code and make sure it returns
what you say it should return.  By encapsulating a set of tests in a test
package, you can quickly check if changes have affected something you didn’t
realize.  It’s a fairly straightforward idea, but if your result is visual, it
becomes difficult to automatically the check.  Still, you would know that the
changed code passed the simpler tests, and you could focus your testing on the
visual tests.

7.3 PORTING ISSUES

Porting code from one platform to another is fairly straightforward in Tcl,
since Tcl does most of the platform specific stuff for you.  However, you may
need to set fonts or options differently for different platforms.  To do so,
you should take advantage of the tcl_platform array and avoid the env variable
(until after you have used tcl_platform to figure out your system).  

1) Use [file join ] instead of simply using the ‘/’ character.  It will save
you time when dealing with “c:\program files”.  The space in “program files”
is the bane of MS-Windows programming.

2) Use Tk’s built-in dialogs instead of writing your own.  This will give a
look and feel that is expected by your users.

7.4 CHANGE FILES

It is often advisable to have a central place for people to go to find out
what has changed in your code.  The file ‘changes’ provides that.  This is
better than at the top of the file (although you may want to put something
there as well), since your changes may span several files.  The change file
should be in chronological order (newest at the bottom) and have a form
similar to:
MM/DD/YYYY (<bug fix or new feature>) <what you did> (<initials or name>)

7.5 WIDGET CREATION COMMENTS / METHODS

When we created the original MDL Comment guide, it was felt that we should
emphasize the tree like structure of a widget.  For example:

# ------------Creating new window---------------------------------------------
#  (tl=top level) (l=label) (f=frame) (c=canvas) (lb=list box) (sb=scroll bar)
#  (b=button) (cb=check button) (rb=radio button) (e=entry) (m=menu)
# $a                  (tl) Main Top level
# $a.label             (l) Explanation label
# $a.top               (f)
# $a.top.x             (l) The x label
# $a.top.y             (l) The y label

# $a.mid               (f)
# $a.mid.new           (l) The new label
# $a.mid.canv          (c) A sample canvas
# $a.bot               (f)
# $a.bot.check        (cb) A check button.
# $a.bot.radio        (rb) A radio button.
# $a.bot.exit          (b) The exit button
# ------------Creating new window---------------------------------------------

Unfortunately this repeats what is in the code (see Section 6.2).  It does
help conceptualize the widget, but is difficult to maintain as the widget
changes.  Also I don’t hold onto the full paths to each widget, but instead
take advantage of the set command.  For example.

set cur [frame $a.top]
  label $cur.x
  label $cur.y
  pack $cur.x $cur.y
set cur [frame $a.mid]
  label $cur.x
  label $cur.y
  pack $cur.x $cur.y
pack $a.top $a.mid

The indentation does help visualize the tree structure, so I definitely
recommend it, but I have found the “Creating new window” stuff only really
helpful for the initial creation of the widget.

Example x-x.  Current MDL Style Guide “template.tcl”

#*****************************************************************************
# <filename>
#
#   <Brief Module Description if this is the primary .tcl file>
#   or <Brief file description, mentioning the primary .tcl file>
#
# History:
#   <mm/yyyy> <Programmer> (<Organization>): <Modification comments>
#
# Notes:
#   <Any features that the programmer thinks would be nice, but hasn't
#    implemented yet, also mentionable features.>
#
#*****************************************************************************
# Global variables:         (This section is intended to describe the Global
variables)
#   <Global array name>
#     (<indices>) : <what it contains>
#   <Any other global variables> : <what they contain>
#*****************************************************************************
set src_dir [file dirname [info script]]
if {[file pathtype $src_dir] != "absolute"} {
  set cur_dir [pwd] ;  cd $src_dir
  set src_dir [pwd] ;  cd $cur_dir
}
package require ...
source [file join $src_dir <secondary .tcl files>]
namespace eval ... {
  namespace export ...
}

#*****************************************************************************
# <namespace>::<Procedure name> --
#
#    Brief description (possibly including who should call it and under what
circumstances)
#
# Arguments:
#    <var name 1> : (optional) Defaults to ..., which means ... expected
type... 
#                              what is its function...
#    <var name 2> :  expected type... what is its function...
#
# Globals:
#   <global variable> : What is it used for..., what does it contain.
#
# History:
#   <mm/yyyy> <Programmer> (<Organization>): <Modification comments>
#
# Notes:
#   <Any features that the programmer thinks would be nice, but hasn't
#    implemented yet, also mentionable features.>
#*****************************************************************************
proc <namespace>::<Procedure name> {var1 var2 {var3 1} var4 etc} {
}

Example x.x.  Original MDL Style Guide “template.tcl” for File Header

#*****************************************************************************
# <filename> :: <Assumed System> (ie Hp 10.0 Tcl/Tk8.0)
#
# Purpose:
#   <Brief Module Description if this is the main .tcl file>
#   or <Brief file description, mentioning the main .tcl file>
#
# Files Needed:
#   Source: <Any .tcl files> <Any packages> <Any libraries> <Any executables>
#   Input: <Any input files (example: a file with default choices)>
#   Output: <Any output files that are created>
#
# Procedures: (P=public) (S=Secretive/private)
# (P) <procs in this file which are intended to be called from outside>
# (S) <procs in this file which are not intended to be called from outside>
#
# History:
#   <mm/dd/yyyy> <Programmer> (<Organization>): <Modification comments>
#
# Notes:
#   <Any features that the programmer thinks would be nice, but hasn't
#    implemented yet, also mentionable features.>
#
#*****************************************************************************
# Global variables:
#   <Global array name>
#     (<indices>) : <what is stored there>
#   <Any other global variables> : <what they store>
#*****************************************************************************

Example x.x.  Original MDL Style Guide “template.tcl” for the Procedure Header

#*****************************************************************************
#  <Proc name>
#
# Purpose:
#   <A Brief description of the purpose of the proc.>
#
# Variables: (I=input)(O=output)(G=global)
#   <Input parameters>: (I) <Purpose>
#   <Output parameters>:(O) <Purpose>
#   <Global variables>: (G) <Purpose>
#
# Files Used: (-- Optional: no need if no files. --)
#   <list of files used by this proc, or variables containing names of files>
#
# Returns:
#   <NULL or what the proc returns>
#
# History:
#   <mm/dd/yyyy>  <Programmer>: <Modification Comments>
#
# Notes:
#   <Any tricky things, or desired improvements>
#*****************************************************************************

       If a new window is being created, put a "Creating New Window Header"
    above the implementation for that new window. (Splitting the "Creating
    New Window Header" is optional (May be used to keep comments closer to
    the implementation).  Use the following line to specify a split.)
# ------------Creating new window--(continued)--------------------------------

# ------------Creating new window---------------------------------------------
#  (tl=top level) (l=label) (f=frame) (c=canvas) (lb=list box) (sb=scroll bar)
#  (b=button) (cb=check button) (rb=radio button) (e=entry) (m=menu)
# <Full path name>     (<widget type>) <contains.>
# ------------Creating new window---------------------------------------------

    Example:

# ------------Creating new window---------------------------------------------
#  (tl=top level) (l=label) (f=frame) (c=canvas) (lb=list box) (sb=scroll bar)
#  (b=button) (cb=check button) (rb=radio button) (e=entry) (m=menu)
# $a                  (tl) Main Top level
# $a.label             (l) Explanation label
# $a.top               (f)
# $a.top.x             (l) The x label
# $a.top.y             (l) The y label
# $a.mid               (f)
# $a.mid.new           (l) The new label
# $a.mid.canv          (c) A sample canvas
# $a.bot               (f)
# $a.bot.check        (cb) A check button.
# $a.bot.radio        (rb) A radio button.
# $a.bot.exit          (b) The exit button
# ------------Creating new window---------------------------------------------

Example x-x.  Simple Example

#*****************************************************************************
# <print.tcl>
#
#     A package intended to ease the creation of console / diagnostic text
#   widgets.  Basic idea: create a text window, and hand over control to this
#   module, calling ns_Print::puts.
#
# History:
#   3/2002 Arthur Taylor (RSIS/MDL): Created
#
# Notes: (example of usage)
#   source p:/ver/verif/print.tcl
#   text .foo -state disabled
#   pack .foo
#   ns_Print::Init .foo
#   ns_Print::puts here
#*****************************************************************************
# Global variables:
#   TextWindow : Contains path to "text" widget or "NULL" to use stdout.
#   ErrorFile  : Filename to use for error messages, or "NULL" to ignore.
#*****************************************************************************
namespace eval ns_Print {
  variable TextWindow NULL
  variable ErrorFile NULL
}

#*****************************************************************************
# ns_Print::Init --
#
#    Initialize the print module.
#
# Arguments:
#    text_path : (optional) Default: NULL, which means use stdout.
#                   otherwise is the "path" to the text widget to put text in.
#    filename  : (optional) Default: NULL, ignore fputs commands.
#                   otherwise name of a valid file to append to.
#
# Globals:
#   TextWindow : Contains path to "text" widget or "NULL" to use stdout.
#   ErrorFile  : Filename to use for error messages, or "NULL" to ignore.
#
# History:
#   3/2002 Arthur Taylor (RSIS/MDL): Created
#
# Notes:
#*****************************************************************************
proc ns_Print::Init {{text_path NULL} {filename NULL}} {
  variable TextWindow $text_path
  variable ErrorFile $filename
  if {$filename != "NULL"} {
    catch {file delete $filename}
    if {! [file exists [file dirname $filename]]} {
      file mkdir [file dirname $filename]
    }
  }
}

#*****************************************************************************
# ns_Print::puts --
#
#    Prints a message to stdout or the Text Window.
#
# Arguments:
#    string    : The message to print.
#    f_newline : (optional) true (1) means add a newline after message,
#                   false (0), don't add newline.
#
# Globals:
#   TextWindow : Contains path to "text" widget or "NULL" to use stdout.
#
# History:
#   3/2002 Arthur Taylor (RSIS/MDL): Created
#
# Notes:
#*****************************************************************************
proc ns_Print::puts {string {f_newline 1}} {
  variable TextWindow
  if {$TextWindow == "NULL"} {
    if {$f_newline} {
      puts $string
    } else {
      puts -nonewline $string
    }
  } else {
    $TextWindow configure -state normal
    $TextWindow insert end $string
    if {$f_newline} {
      $TextWindow insert end "\n"
    }
    $TextWindow configure -state disabled
  }
  $TextWindow see end
  update
}

#*****************************************************************************
# ns_Print::fputs --
#
#    Appends a message to ErrorFile (if not NULL)
#
# Arguments:
#    string    : The message to print.
#    f_newline : (optional) true (1) means add a newline after message,
#                   false (0), don't add newline.
#
# Globals:
#   ErrorFile  : Filename to use for error messages, or "NULL" to ignore.
#
# History:
#   3/2002 Arthur Taylor (RSIS/MDL): Created
#
# Notes:
#*****************************************************************************
proc ns_Print::fputs {string {f_newline 1}} {
  variable ErrorFile
  if {$ErrorFile != "NULL"} {

    set fp [open $ErrorFile a]
    if {$f_newline} {
      ::puts $fp $string
    } else {
      ::puts $fp -nonewline $string
    }
    close $fp
  }
}


